
1. De�niciones en forma BNF
< > Símbolos no terminales. Declaraciones u objetos declarados
::= Operador "se de�ne como"
| Operador disyuntivo "ó"

[ ] Declaración/parte opcional
{ } Declaración/parte repetitiva (0..∞ veces)
α..ω Rangos de caracteres
Otros Símbolos y secuencias terminales. Literales
' ' Literales que incluyen símbolos especiales

2. Notación Algorítmica
2.1. De�niciones básicas
<letra> ::= A..Z | a..z
<dígito> ::= 0..9

<símbolo> ::= _ | % | & | : | , | ; | = | ( | etc...
<caracter-nombre> ::= <letra> | <dígito> | _
<caracter> ::= <letra> | <dígito> | <símbolo>

2.2. Valores y literales
<valor> ::= <valor-numérico> | <valor-caracter> | <valor-cadena>

| <valor-lógico>

<valor-numérico> ::= [+ | −] <digito> {<digito>}
[ . {<digito>} ] [E [+ | −] <digito> {<digito>} ]

<valor-caracter> ::= ′ <caracter> ′

<valor-cadena> ::= ′ {<caracter>} ′

<valor-lógico> ::= V | F

2.3. Expresiones
<expresión> ::= <exp-numérica> | <exp-caracter> | <exp-cadena>

| <exp-relacional> | <exp-lógica>

<exp-numérica> ::= <exp-numérica> <operador-numérico> <exp-numérica>
| ( <exp-numérica> ) | −<exp-numérica> | <nombre-objeto>
| <valor-numérico>

<operador-numérico> ::= + | − | * | / | div | mod
<exp-caracter> ::= <nombre-objeto> | <valor-caracter>

<exp-cadena> ::= <exp-cadena> +<exp-cadena> | <nombre-objeto>
| <valor-cadena>

<exp-relacional> ::= <expresión> <operador-relacional> <expresión>

<operador-relacional> ::= > | >= | < | <= | = | <>

<exp-lógica> ::= <exp-relacional> | <operador-lógico1> <exp-lógica>
| <exp-lógica> <operador-lógico2> <exp-lógica>
| ( <exp-lógica> ) | <nombre-objeto> | <valor-lógico>

<operador-lógico1> ::= NOT

<operador-lógico2> ::= AND | OR

2.4. Declaración de objetos
<declaración-objeto> ::= <decl-obj-simple> | <decl-obj-compuesto>

| <decl-vector> | <decl-matriz> | <decl-�chero>

2.4.1. Declaración de bjetos simples
<decl-obj-simple> ::= <decl-constante> | <decl-variable>

<decl-constante> ::= <nombre-constante> = <valor>

<decl-variable> ::= <nombre-variable> : <nombre-tipo-simple>

<nombre-tipo-simple> ::= Numérico | Caracter | Cadena | Lógico

2.4.2. Declaración de objetos compuestos
<decl-obj-compuesto> ::= <decl-tipo-compuesto> | <decl-variable-compuesta>

<decl-tipo-compuesto> ::=
REGISTRO <nombre-tipo-compuesto>
<decl-objeto>
{ <decl-objeto> }
FIN-REGISTRO

<decl-variable-compuesta> ::= <nombre-variable> : <nombre-tipo-compuesto>

2.4.3. Declaración de objetos vector y matriz
<tipo> ::= <nombre-tipo-simple> | <nombre-tipo-compuesto>

<rango> ::= <valor-numérico> .. <valor-numérico>

<decl-vector> ::= <nombre-variable> : Vector '[' <rango> ']' de <tipo>

<decl-matriz> ::= <nombre-variable> : Vector '[' <rango> ']' de
{ Vector '[' <rango> ']' de } <tipo>

| <nombre-variable> : Vector '[' <rango> ']' { '[' <rango> ']'
} de <tipo>

2.5. Nombres y uso de objetos
<nombre-objeto> ::= <nombre-constante> | <nombre-variable>

| <nombre-elemento-var-comp>

| <nombre-elemento-vector>
| <nombre-elemento-matriz>

<nombre-constante> ::= <letra> { <caracter-nombre> }
<nombre-variable> ::= <letra> { <caracter-nombre> }
<nombre-elemento-var-comp> ::= <nombre-variable> { • <nombre-variable>

}
<nombre-elemento-vector> ::= <nombre-variable>'['<valor-numérico>']'

<nombre-elemento-matriz> ::=
<nombre-variable>'['<valor-numérico>']' { '['<valor-numérico>']'
}

2.6. Objetos �chero
<nombre-�chero> ::= <letra> { <caracter-nombre> }
<decl-�chero> ::= <nombre-�chero> : Fichero de <tipo>

<operación-�chero> ::= <apertura-�chero> | <cerrado-�chero>
| <lectura-�chero> | <escritura-�chero>

<apertura-�chero> ::= Abrir ( <nombre-�chero> )

<lectura-�chero> ::= Leer ( <nombre-�chero> , <nombre-objeto> )

<escritura-�chero> ::= Escribir ( <nombre-�chero> , <nombre-objeto>
)

<cerrado-�chero> ::= Cerrar ( <nombre-�chero> )

2.7. Acciones
<acción> ::= <acción-elemental> | <composición-secuencial-acciones>

| <esquema-condicional> | <esquema-repetitivo>

<acción-elemental> ::= <asignación-interna> | <asignación-externa>
| <escritura> | <operación-�chero>

<asignación-interna> ::= <nombre-variable> := <valor>

<asignación-externa> ::= Leer <nombre-objeto> { , <nombre-objeto>
}

<escritura> ::= Escribir <nombre-objeto> { , <nombre-objeto> }

2.8. Esquemas de programación estructurada
2.8.1. Esquema composición secuencial
<composición-secuencial-acciones> ::= <acción> { ; <acción> } [ ; ]

2.8.2. Esquema condicional
<esquema-condicional> ::=

<esquema-cond-1> | <esquema-cond-2> | <esquema-cond-generalizado>

<esquema-cond-1> ::=
SI <exp-lógica> ENTONCES
<acción>

FIN-SI

<esquema-cond-2> ::=
SI <exp-lógica> ENTONCES
<acción>

SINO
<acción>

FIN-SI

<esquema-cond-generalizado> ::=
SEGUN <expresión> HACER
<caso> { ; <caso> }
[ ; <otro-caso> ]

FIN-SEGUN

<caso> ::= <valor> { , <valor> } : <acción>

<otro-caso> ::= otros : <acción>

2.8.3. Esquema repetitivo
<esquema-repetitivo> ::= <esquema-mientras> | <esquema-repetir> | <esquema-

para>

<esquema-mientras> ::=
MIENTRAS <exp-lógica> HACER
<acción>

FIN-MIENTRAS

<esquema-repetir> ::=
REPETIR
<acción>

HASTA QUE <exp-lógica>

<esquema-para> ::=
PARA <nombre-objeto> DESDE <valor-numérico> HASTA <valor-numérico>
[ INCREMENTO <valor-numérico> ]

<acción>

FIN-PARA

2.9. Algoritmo
<algoritmo> ::=

Algoritmo <nombre-algoritmo> [ ( <lista-parámetros> ) ] es
<sección-declaraciones>
inicio

<acción>

fin

<nombre-algorítmo> ::= <letra> { <caracter-nombre> }
<lista-parámetros> ::= <nombre-variable> { , <nombre-variable> }
<sección-declaraciones> ::= { <declaración-objeto> }

2.10. Documentación
<comentario> ::= '{' { <caracter> } '}'

